# The two highest coefficients of a symmetric power of a second order operator.

by:
Mark van Hoeij
Florida State University
May 2001.

```
> with(DEtools):
  _Envdiffopdomain:=[Dx,x]: # means d/dx is denoted by Dx
```

## (n-1)'th symmetric power of Dx^2 + a*Dx

L = Dx^2 + a*Dx
Basis of solutions of L:   1, y.
Then y' is a solution of Dx+a.
Now Y := 1/y' is a solution of Dx-a so Y'=a*Y.

Let Ln = symmetric_power(L,n-1) = Dx^n + An * Dx^(n-1) + Bn * Dx^(n-2) + ...  where the dots stand for lower order terms.
Solutions of Ln:  1,y,y^2,..,y^(n-1).
Define L_n as (d/dy)^n. Solutions of L_n are the same as the solutions of Ln.
d/dy = 1/y' * d/dx = Y*d/dx
L_n = (Y*d/dx)^n = Y^n * Dx^n + ...
Since L_n has the same solutions as Ln we have L_n = Y^n * Ln  and so:
L_n = (Y*d/dx)^n = Y^n * Dx^n + Y^n * An * Dx^(n-1) + Y^n * Bn * Dx^(n-1) + ...

Now write L_n = sum(c[i,n]*Dx^i, i=0..n).  Then sum(c[i,n+1]*Dx^i, i=0..n+1) = L_(n+1) = Y*Dx*L_n = Y*sum( (Dx*c[i,n])*Dx^i,i=0..n) = Y*sum( (c[i,n]*Dx + c[i,n]')*Dx^i,i=0..n) = Y*sum( (c[i-1,n]+c[i,n]')*Dx^i,i=0..n+1) and thus:
   c[i,n+1] = Y * (c[i-1,n] + c[i,n]')
Now c[0,n]=0 and c[n,n]=Y^n  for all n>0. Let d[n]=c[n-1,n]. Then d[1]=0 and
   d[n+1] = Y * (d[n] + c[n,n]') = Y * (d[n] + (Y^n)') = Y * (d[n] + a*n*Y^n)
It is easy to solve this recurrence:
   d[n] = n*(n-1)/2 * a * Y^n
Hence:
   An = n*(n-1)/2 * a.

Now define e[n] = c[n-2,n]. Then e[1]=e[2]=0 and
   e[n+1] = Y * (c[n-2,n] + c[n-1,n]') = Y * (e[n] + d[n]') = Y * (e[n] + n*(n-1)/2 * (a*Y^n)')
And (a*Y^n)' = a'*Y^n + a*(Y^n)' = Y^n * (a' + n*a^2)  so

e[n+1] = Y*(e[n] + n*(n-1)/2*Y^n*(a’+n*a^2))
The solution of the recurrence is:
e[n]=Y^n * sum(i*(i-1)/2*(a’ + i*a^2), i=0..n-1)

```
> e[n]/Y^n = factor(sum(i*(i-1)/2 * (ap + i*a^2),i=0..n-1));
```

$$\frac{e_n}{Y^n} = \frac{1}{24}\, n\,(n-1)\,(n-2)\,(3\,n\,a^2 - a^2 + 4\,ap)$$

Conclusion:
Bn = 1/24 * n*(n-1)*(n-2) * (4*a’ + (3*n-1)*a^2)

# (Dx+b)^n

Now define L_n as (Dx+b)^n. Write L_n = sum(c[i,n]*Dx^i,i=0..n). Then:
sum(c[i,n+1]*Dx^i, i=0..n+1) = L_(n+1) = (Dx+b)*L_n = sum( (Dx+b)*c[i,n]*Dx^i, i=0..n) = sum( (b*c[i,n] + c[i,n]’ + c[i,n]*Dx)*Dx^i, i=0..n) = sum( (b*c[i,n]+c[i,n]’+c[i-1,n])*Dx^i,i=0..n) and so:
c[i,n+1] = b*c[i,n] + c[i,n]’ + c[i-1,n]
Now: c[0,1]=b and c[n,n]=1 for all n. Define d[n]=c[n-1,n] so d[1]=b and
d[n+1] = b*1 + 1’ + d[n] = b + d[n]
Hence:
d[n] = n*b
Now define e[n]=c[n-2,n] so e[1]=0 and
e[n+1] = b*c[n-1,n] + c[n-1,n]’ + c[n-2,n] = b*d[n] + d[n]’ + e[n] = n*b^2 + n*b’ + e[n]
and so:
e[n] = sum(i,i=0..n-1)*(b^2+b’) = n*(n-1)/2 * (b^2 + b’)
So:
(Dx+b)^n = Dx^n + n*b * Dx^(n-1) + n*(n-1)/2 * (b^2+b’) * Dx^(n-2) + ...

# (n-1)’th symmetric power of Dx^2 + (a+2*b)*Dx + a*b+b’+b^2

If L=sum(c[i]*Dx^i, i=0..n) with c[n]=1 then:

symmetric_product(L, Dx+b) = sum(c[i]*(Dx+b)^i, i=0..n)

and we can evaluate the highest order terms of each (Dx+b)^i with the formula from the previous section.

```
> L := symmetric_product(Dx^2+a(x)*Dx,  Dx+b(x));
```

$$L := Dx^2 + (2\,\mathrm{b}(x) + \mathrm{a}(x))\,Dx + \mathrm{a}(x)\,\mathrm{b}(x) + \left(\frac{\partial}{\partial x}\mathrm{b}(x)\right) + \mathrm{b}(x)^2$$

Let L = symmetric_product(Dx^2 + a*Dx,  Dx+b)

L = (Dx+b)^2 + a*(Dx+b) = Dx^2 + (a+2*b)*Dx + a*b + b' + b^2

symmetric_power(L, n-1)
 = symmetric_product(
 symmetric_power(Dx^2 + a*Dx, n-1),
 symmetric_power(Dx+b, n-1) )
 = symmetric_product(
 Dx^n + An*Dx^(n-1) + Bn*Dx^(n-2) + ... ,
 Dx + b1 ) where b1 = (n-1)*b
 = (Dx + b1)^n + An * (Dx + b1)^(n-1) + Bn * (Dx + b1)^(n-2) + ...
 = Dx^n + n*b1 * Dx^(n-1) + n*(n-1)/2 * (b1^2+b1') * Dx^(n-2) + ...
 + An * (Dx^(n-1) + (n-1)*b1 * Dx^(n-2) + ... )
 + Bn * (Dx^(n-2) + ... )

```
> b1 := (n-1)*b(x):
  An := n*(n-1)/2 * a(x):
  Bn := 1/24*n*(n-1)*(n-2)*(4*diff(a(x),x)+(3*n-1)*a(x)^2):
  sp :=
  Dx^n + n*b1*Dx^(n-1)+n*(n-1)/2*(b1^2+diff(b1,x))*Dx^(n-2)
  + An * ( Dx^(n-1) + (n-1)*b1*Dx^(n-2) )
  + Bn * Dx^(n-2);
```

$$sp := Dx^n + n\,(n-1)\,\mathrm{b}(x)\,Dx^{(n-1)} + \frac{1}{2}\,n\,(n-1)\left((n-1)^2\,\mathrm{b}(x)^2 + (n-1)\left(\frac{\partial}{\partial x}\,\mathrm{b}(x)\right)\right)Dx^{(n-2)}$$
$$+\frac{1}{2}\,n\,(n-1)\,\mathrm{a}(x)\,(Dx^{(n-1)} + (n-1)^2\,\mathrm{b}(x)\,Dx^{(n-2)})$$
$$+\frac{1}{24}\,n\,(n-1)\,(n-2)\left(4\left(\frac{\partial}{\partial x}\,\mathrm{a}(x)\right) + (3\,n-1)\,\mathrm{a}(x)^2\right)Dx^{(n-2)}$$

```
> CF_n_1 := factor(coeff(sp,Dx^(n-1)));
```

$$CF\_n\_1 := \frac{1}{2}\,n\,(2\,\mathrm{b}(x) + \mathrm{a}(x))\,(n-1)$$

```
> CF_n_2 := collect(coeff(sp,Dx^(n-2)), {a(x),b(x)},factor);
```

$$CF\_n\_2 := \frac{1}{24}\,n\,(n-1)\,(n-2)\,(3\,n-1)\,\mathrm{a}(x)^2 + \frac{1}{2}\,n\,(n-1)^3\,\mathrm{a}(x)\,\mathrm{b}(x) + \frac{1}{2}\,n\,(n-1)^3\,\mathrm{b}(x)^2$$
$$+\frac{1}{6}\,n\,(n-1)\left(3\,n\left(\frac{\partial}{\partial x}\,\mathrm{b}(x)\right) + n\left(\frac{\partial}{\partial x}\,\mathrm{a}(x)\right) - 3\left(\frac{\partial}{\partial x}\,\mathrm{b}(x)\right) - 2\left(\frac{\partial}{\partial x}\,\mathrm{a}(x)\right)\right)$$

We can now write
 L = Dx^2 + A*Dx + B
where

```
> A(x)=coeff(L,Dx,1);
```

$$\mathrm{A}(x) = 2\,\mathrm{b}(x) + \mathrm{a}(x)$$

```
> B(x)=coeff(L,Dx,0);
```

$$B(x) = a(x)\,b(x) + \left(\frac{\partial}{\partial x} b(x)\right) + b(x)^2$$

```
> S:=dsolve({%,%%}, {a(x),b(x)});
```

$$S := \left[\{\frac{\partial}{\partial x} b(x) = B(x) - b(x)\,A(x) + b(x)^2\},\ \{a(x) = A(x) - 2\,b(x)\}\right]$$

```
> map(collect,subs(op(S),expand(subs(op(S),[CF_n_1,CF_n_2]))),
  {A(x),B(x)}, factor);
```

$$\left[\frac{1}{2}\,n\,A(x)\,(n-1),\right.$$
$$\left.\frac{1}{24}\,n\,(n-1)\,(n-2)\,(3\,n-1)\,A(x)^2 + \frac{1}{6}\,n\,(n-1)\,(n+1)\,B(x) + \frac{1}{6}\,n\left(\frac{\partial}{\partial x} A(x)\right)(n-1)\,(n-2)\right]$$

**We conclude the following:**

```
>
```

# Proposition: **If**

**L = Dx^2 + A*Dx + B**

**Then**

**symmetric_power(L, n-1) = Dx^n   + An * Dx^(n-1) + Bn * Dx^(n-2) + ...**

**where**

**An = 1/2*n*(n-1) * A**

**and**

**Bn = 1/24*n*(n-1)*(n-2)*(3*n-1)*A^2 + 1/6*(n-1)*n*(n+1)*B+1/6*n*(n-1)*(n-2)*A'**

We proved this for the case  where A=a+2*b,  B=a*b+b'+b^2 for some functions a,b. Since all A,B can be written in this form, it follows that the proposition is true in general.
Note that it is trivial to verify for any particular value of n that the formula is correct, for example, if you want to verify the case n=4 then just type:

n:=4; with(DEtools):  symmetric_power(Dx^2 + a(x)*Dx + b(x), n-1, [Dx,x]);

in Maple. However, this approach does not provide a formula for general n.

# Application: Detection of symmetric powers of second order operators.

**Input:**   L = Dx^n + An * Dx^(n-1) + Bn * Dx^(n-2) + ...

**Output is either:**

"L is not a symmetric power of a 2nd order operator"

**or:**

A,B  such that L is (n-1)-th symmetric power of Dx^2 + A*Dx + B

# Algorithm:

Step 1: A := 2/(n*An/(n*(n-1))
Step 2: B := 6/((n-1)*n*(n+1)) * Bn - (n-2)/(n+1) * Ap - (3*n-1)*(n-2)/(n^2*(n-1)^2*(n+1)) *An^2
Step 3: Compute (n-1)'th symmetric power of Dx^2+A*Dx+B with the algorithm of Bronstein/Mulders/Weil.
Step 4: Test if L equals this symmetric power. If so, then output A,B, otherwise L is not a symmetric power of a 2nd order operator.

**Implementation** (in MapleV 5).
Note 1: m=n-1.
Note 2: The formulas on which the following algorithm is based were computed by the author of this document in 1997 but the computation was not written down until now. The implementation was done by G. Labahn.

```
proc(A, x)
local i, a, DF, newA, b, ans, m, compare, y;
option `Copyright (c) 1997 George Labahn. All rights reserved.`;
   m := nops(A) - 2;
   newA := [seq(A[i]/A[m + 2], i = 1 .. m + 2)];   # Make L monic.
   a := 2*newA[m + 1]/(m*(m + 1));
   b := normal(1/4*(-3*m^4*a^2 - 2*m^3*a^2 - 4*m^3*diff(a, x) + 3*m^2*a^2
        + 2*m*a^2 + 4*m*diff(a, x) + 24*newA[m])/(m*(m^2 + 3*m + 2)));
   compare := DEtools[symmetric_power](DF^2 + a*DF + b, m, [DF, x]);
   for i from 0 to m do
      if simplify(coeff(compare, DF, i) - newA[i + 1]) <> 0 then
         RETURN([])
      fi
   od;
   ans := readlib('`dsolve/diffeq/linearODE`')(b*y + a*_.y[1] + _.y[2], x, y);
   if ans <> [] and nops(ans) = 2 and not has(ans, 'DESol') then
      # ans[1] and ans[2] are the solutions of the 2nd order operator, so
      # their products are the solutions of L:
      RETURN([seq(ans[1]^i*ans[2]^(m - i), i = 0 .. m)])
   fi;
   []
end
```

# Reference:

Bronstein, Mulders, Weil. "On Symmetric Powers of Differential Operators", ISSAC'1997.